

## Visualization of Self-Reported Alcohol Use Data: A Timeline Followback Dashboard using Heatmaps, Time Series Plots, and HTML Hover-Text

Daniel Fuhrmann, University of Wisconsin-Milwaukee, Milwaukee, WI  
Michael Fendrich, University of Connecticut, West Hartford, CT  
Lisa Berger, University of Wisconsin-Milwaukee, Milwaukee, WI

### ABSTRACT

In research studies of substance use, a widely employed computerized technique for assessing self-reported alcohol consumption is the Timeline Followback (TLFB) interview. The TLFB data for the present study were collected from a random sample of 606 undergraduate students from a large, public urban university. Study participants reported their daily alcohol use measured in US standard drinks for a pre-specified time horizon (90 calendar days prior to the interview date). In order to increase the accuracy of the reports, TLFB assessors provided the participants with reference guides on the definition of a US standard drink, encouraged them to think about any established drinking behavior patterns, and offered TLFB calendars listing local events and participant personal events as anchors to aid with recall. Typically, summary measures of alcohol use – such as total number of drinks, average drinks per drinking day, etc. – are constructed from these kinds of data.

The purpose of this paper is to demonstrate a novel way to visualize the interplay between reported alcohol use and social context. With the goal of data exploration in mind, we construct a dashboard using SAS/GRAPH® procedures that displays (1) the individual-level drinking trajectories over time (by calendar day) with each day color-coded based on drinking level and reported personal or social events; (2) time series plots, box plots, and frequency tables to summarize drinking and event data across participants; (3) standard descriptive statistics across time for each participant (Sobell & Sobell's "Drinking Picture"); and (4) a global study summary table.

### KEYWORDS

Longitudinal drinking data, social context of drinking behavior, drinking patterns, data visualization, heat map, boxplots, time series plots, annotation, PROC GPLOT, PROC SGPLOT, PROC BOXPLOT, PROC SQL

### INTRODUCTION

Heavy drinking is associated with alcohol-related problems among college students. A recent study by Beck et al. (2008) examined the relationship between social context of drinking and alcohol problems in a large sample of college students. The authors found significant associations between social context and alcohol use disorders, other alcohol-related problems, and depression. Understanding the social circumstances of alcohol use behavior is important as it provides information about the situational factors that may influence or interact with individuals' drinking patterns. While many studies have explored the relationship between drinking and alcohol-related consequences, the longitudinal data about alcohol use are often reported using composite measures of drinking such as drinking frequency (e.g. percent days abstinent) or drinking intensity (average drinks per drinking day, percent heavy drinking days). This approach leads to a loss of information, however, as detailed drinking information is aggregated into summary statistics.

This limitation was recognized by Krentzman et al. (2010) who were the first to construct heat maps as a means to evaluate longitudinal drinking outcomes. Utilizing data from 364 alcoholics who were followed over a period of 2.5 years (eight 90-day TLFB periods), the authors constructed heat maps in Microsoft Excel (Windows Office 2007) using conditional formatting to classify drinking patterns. The approach was innovative and yielded valuable insights about the drinking patterns that were present in their study sample, but the heat map cell values were also constructed using aggregation (mean drinks per drinking day in each 90-day period).

The purpose of this paper is to demonstrate a novel way to visualize the interplay between daily reported alcohol use and social context. With the goal of data exploration in mind, we create several charts, plots, and a large heat map that could ultimately form a dashboard to effectively visualize longitudinal drinking data. Several SAS/GRAPH® procedures are employed to display individual-level drinking trajectories over time (by calendar day) with each day color-coded based on drinking level and reported personal or social events. PROC BOXPLOT, PROC SGPLOT, and PROC FREQ are used to create time series plots, box plots, and frequency tables to summarize drinking and event

data across participants. PROC GPLOT with annotated markers for participant ID values and social events is used to create a large heat map in HTML and PNG output formats to display individual drinking levels by day over the whole study period. Selected descriptive statistics across time for each participant are constructed for illustration purposes and a global study summary table is created with PROC SQL to complete the picture.

## METHODS

The data for the present study were obtained from a random sample of 606 undergraduate students from a large, public urban university. Survey instruments like the calendar-based Timeline Follow Back method (Sobell & Sobell, 1992; Sobell, Brown, Leo, & Sobell, 1996) were used to collect self-reported data on alcohol drinking patterns and consumption levels during the 90-day time period preceding the interview day. Based on responses to the structured diagnostic interview M.I.N.I. (Sheehan et al., 2009), study participants were classified for the presence (or absence) of alcohol abuse or dependence based on DSM-IV criteria (American Psychiatric Association, 2000). Hair and fingernail specimens were collected from the majority of study participants and tested for their concentration of the alcohol biomarker ethyl glucuronide (EtG) (see Jones et al., 2012, for more detail).

Out of the many measures that were collected in this study, we focus on the TLFB alcohol use data. At the time of the interview, study participants were presented with an event calendar that was pre-filled with events that were deemed relevant to a college-age population. These event calendars are frequently being used to aid with recall of alcohol use behavior. Study interviewers entered numbers corresponding to the participant-reported values of standard alcohol drinks consumed on each day into an electronic template of the 90-day period prior to the respective participant's interview date. Participants also had the opportunity to add additional social events into their calendars above and beyond those that were pre-filled by study staff.

For illustrative purposes, this paper is based on the same study sample that was analyzed in Berger et al. (2013). From the original 606 observations, a matched sample of n=447 based on hair and fingernail specimen criteria and non-missing drinking behavior is retained for the present analyses and visualizations.

## Examples of Social Events

Prior to data collection, research staff created lists of social events for the geographical area in which recruitment for the study was going to take place. These lists were transferred onto calendar templates used by study interviewers for the TLFB alcohol use self-report. Sources for these events were academic calendars and geographically relevant websites. An example of a pre-populated event calendar that was used during the study is displayed in Figure 1 below.

# May 2010

Sun	Mon	Tue	Wed	Thu	Fri	Sat
						1 -Mifflin Street Block Party (Madison)
2 -Milwaukee Bucks vs. Atlanta Hawks NBA Playoffs: Game 7 (L)	3	4	5 -Cinco de Mayo	6	7	8
9 -Mother's Day	10	11	12	13	14	15 -UWM Final Exams Begin
16 -UWM Graduation	17 -UWM Final Exams	18 -UWM Final Exams	19 -UWM Final Exams	20 -UWM Final Exams	21 -UWM Final Exams	22 -UWM Final Exams End -Goodland Reggae Fest
23	24	25	26	27	28	29 -Family Kite Festival
30 -Family Kite Festival	31 -Memorial Day					

Figure 1: Sample calendar of pre-filled social events used as recall aid for TLFB data collection

Broadly speaking, the events could initially be categorized into the following groups:

- Academic/School
- Arts & Entertainment
- Community
- Eat & Drink
- Holidays & Festivals
- Music
- Sports

## ANALYSIS

Data were initially stored in wide format, meaning that each of the 447 observations was represented by one row in the input data set. Time-variant alcohol use data and corresponding event reports were stored in two sets of 90 columns each called drinks\_1-drinks\_90 and events\_1-events\_90. Time-invariant information, such as participant age at interview, sex, and biomarker specimen characteristics (EtG), among others, were stored in individual columns in the dataset. While the value set for reported standard drinks was bound to missing, zero, or positive continuous values, the number of events that students reported in the study exceeded 1,900 unique values! However, more than 90% of the events were comprised of the following:

The FREQ Procedure				
Event	Frequency	Percent	Cumulative Frequency	Cumulative Percent
	29003	72.09	29003	72.09
Summerfest	3232	8.03	32235	80.13
UWM Final Exams	1187	2.95	33422	83.08
Fourth of July	349	0.87	33771	83.94
Birthday	325	0.81	34096	84.75
Father's Day	318	0.79	34414	85.54
NBA Game 7	295	0.73	34709	86.28
NBA Game 1	226	0.56	34935	86.84
UWM Summer	218	0.54	35153	87.38
Memorial Day	206	0.51	35359	87.89
Classes End	202	0.50	35561	88.39
Work	195	0.48	35756	88.88
Graduation Party	179	0.44	35935	89.32
UWM Fall	160	0.40	36095	89.72
Labor Day	153	0.38	36248	90.10

Table 1: Top 15 event listings out of 40,230 possible event cells (447 \* 90 = 40,230)

In an attempt to match the individual event reports to higher-level categories as commonly seen on social media and event websites, the first author and a research assistant grouped the unique event values into the following categories:

The FREQ Procedure				
Event_Cat	Frequency	Percent	Cumulative Frequency	Cumulative Percent
	29003	72.09	29003	72.09
Holidays and Festivals	4320	10.74	33323	82.83
School	2075	5.16	35398	87.99
Sports	950	2.36	36348	90.35
Party/Birthday (Non-family)	904	2.25	37252	92.60
Travel/Vacation	710	1.76	37962	94.36
Family	707	1.76	38669	96.12
Unclassified	523	1.30	39192	97.42
Eat and Drink	415	1.03	39607	98.45
Employment	306	0.76	39913	99.21
Music	105	0.26	40018	99.47
Arts and Entertainment	82	0.20	40100	99.68
Community	62	0.15	40162	99.83
Moving/Relocating	47	0.12	40209	99.95
Medical	19	0.05	40228	100.00
Party/Birthday(non-family)	1	0.00	40229	100.00
Travel / Vacation	1	0.00	40230	100.00

Table 2: Event categories based on ad-hoc sorting

We were still left with 16 event categories after clustering due to the complexity of events reported.<sup>1</sup> It is also worth noting that only about 18% of events reported were so-called pre-populated events in the TLFB calendar. The remaining 82% of event cells were either unique contributions by student participants, or, as the vast majority, missing values. Since such a large number of event categories would still be nearly impossible to incorporate into a visualization of drinking behavior, we will focus on whether an event was reported on any given day or not for subsequent analyses and graphics.

## Data Transformation from WIDE into LONG form

Data can be converted from wide into long format using PROC TRANSPOSE or via the DATA step. The latter is illustrated below for the time-variant drinking and event variables as well as a select few time-invariant variables:

```
data events3;
  set events2;
  array events_new {90} $ events_new1-events_new90;
  array drinks {90} drinks_1-drinks_90;
  array eventscat {90} $ eventcat1-eventcat90;
  array event_prepop {90} event_prepop1-event_prepop90;
  do TLFb_Day = 1 to 90;
    StDrinks=drinks{TLFB_Day};
    Event=events_new{TLFB_Day};
    Event_Cat=eventscat{TLFB_Day};
    Event_Prep=event_prepop{TLFB_Day};
    Day=91-TLFB_Day;
    Cal_Day=calendarend_new+1-TLFB_Day;
    hair_etg=hair_res;
    nail_etg=nail_res;
    hair_pstv=hair_pstv;
    nail_pstv=nail_pstv;
    hair_gte30=hair_gte30;
    nail_gte20=nail_gte20;
    output;
  end;
  keep id StDrinks Event [...];
  format Cal_Day mmddyy10.;
run;
```

The transposed data set **events3** had 40,230 rows with 90 rows for each of the 447 participants in this sample.

## Time Series Plots using PROC SGPLOT

A standard approach to visualize longitudinal data is the time series plot. It can easily be created using the SGPLOT procedure in SAS/GRAPH® (SAS, 2014):

```
goptions reset=all;
proc sgplot data=events3;
  series y=StDrinks_hair_gte301 x=Day / group=id transparency=0.5
    lineattrs=(color=red pattern=1);
  series y=StDrinks_hair_lt301 x=Day / group=id transparency=0.5
    lineattrs=(color=cyan pattern=1);
  pbspline y=StDrinks x=Day / group=hair_gte302 nomarkers smooth=50 nknots=5
    lineattrs=(thickness=3) name="hairetg";
  xaxis values=(0 to 90 by 1) label='Time in Days';
  yaxis values=(0 to 35 by 5) label='Standard Drinks per Day';
  keylegend "hairetg";
  format hair_gte30 hair30etg.;
  title 'Individual Drinking Profiles with Hair EtG Subgroups';
run;
quit;
```

---

<sup>1</sup> Arguably, our ad-hoc sorting scheme may be outperformed by other sorting algorithms such as pile sorting. However, we believe that since the study research staff used event websites to create lists of event suggestions for the TLFB calendar, it would be reasonable to retro-fit the reported events to the same higher-level categories found on those websites.

In the example above, alcohol use for two groups of students is being investigated. The first group had a hair EtG biomarker value of at least 30 pg/mg, which is considered to indicate chronic excessive alcohol consumption according to the Society of Hair Testing consensus guidelines (SOHT, 2011). The second group fell below that cut-off. ❶ In order to display the data using different colors for each group, separate vectors of standard drink counts by biomarker test status were created and plotted using unique SERIES statements in PROC SGPLOT.

❷ Fitted penalized B-spline curves were plotted for each biomarker group to illustrate average drinking differences across both groups over time.

The output produced by this PROC step is displayed below in Figure 2.




Figure 2: Individual drinking profiles over 90-day TLFB with penalized B-splines for biomarker groups

It is obvious that this kind of spaghetti plot has its limitations as the number of study subjects increases. Even if we attempted to label the individual trajectories, the 447 ID values would likely overlap and become unreadable. The heat map approach below will allow for a more detailed display of individual drinking behavior over time, but it will admittedly use a much larger graph area in HTML format.

## DISPLAYING DRINKING PATTERNS IN A HEAT MAP USING ANNOTATIONS

In this section, we demonstrate the use of PROC SQL, annotated output, and PROC GPLOT to create a heat map of individual-level drinking patterns over the entire study time period.

We start out by defining a macro variable *name* and assigning it the character string that will be used to name the final .htm file. We also specify the path for our .htm file to be saved to using a FILENAME statement.

### Data Preparation

```
%let name=drinking_heatmap;
filename odsout '.';
```

Next, we use PROC SQL to get one row per ID value in a new table called **start\_dates** that contains the start date of the 90-day TLFB period for each of the 447 participants:

```
proc sql;
create table start_dates as
select unique id, min(cal_day) as start_day
from events3
group by id;
quit; run;
```

The newly created table **start\_dates** is then sorted by *start\_day* and *id* in ascending order:

```
proc sort data=start_dates out=start_dates;
by start_day id;
run;
```

Using the automatic variable *\_n\_*, we assign values to the variable *y\_axis\_number* based on the number of iterations of the DATA step and merge those numbers back via a SQL left join to the original data set:

```
data start_dates; set start_dates;
y_axis_number=_n_;
run;
```

```
proc sql;
create table events3 as
select unique events3.*, start_dates.y_axis_number
from events3 left join start_dates
on events3.id=start_dates.id;
quit; run;
```

As a next step, we create a macro variable *max\_y* that will store the largest value of the variable *y\_axis\_number*. This maximum value will be used later on when we declare the height (i.e. the row count of drinking patterns to display) of the vertical axis in our heat map.

```
proc sql;
select max(y_axis_number) into :max_y from events3;
quit; run;
```

## Annotations

Now that the raw data preparation is complete, we turn our attention to the annotated output. Below, we create annotated markers with pre-defined colors from the RGB color scheme in SAS/GRAPH for specific drinking levels. The marker is white when zero drinks were reported for any given day and become progressively darker in discrete steps as the number of standard drinks increases.<sup>2</sup> We chose a square marker for the display of a single day's worth of information, including drinking quantity and whether an event was cross-listed or not.

```
data anno_markers;
set events3;
length function $8 color $12 style $35 text $50;
xsys='2'; ysys='2'; hsys='d'; when='a';
x=cal_day; y=y_axis_number;
function='label'; style='albany amt/unicode'; position='5'; size=8;
/* set marker fill color using RGB color scheme here */
if stdrinks=0 then color='white';
else if stdrinks<=1 then color='cxffffb2';
else if stdrinks<=2 then color='cxfed976';
else if stdrinks<=3 then color='cxfeb24c';
else if stdrinks<=4 then color='cxfd8d3c';
else if stdrinks<=5 then color='cxf03b20';
else if stdrinks>5 then color='cxbd0026';
else color='blue'; /* this is a fail-safe, but no data should match this last 'else'
statement since we are working with non-missing data here - a solid blue marker in the
output would indicate a problem */
text='25a0'x; output; /* this is a solid-colored square marker, of the 'albany
amt/unicode' font */
html='';
```

<sup>2</sup> Note that these markers are not gender-specific.

```

/* marker outline color */
if trim(left(event)) ^= '' then color='cx0276FD'; /* special 'event' days get a blue
border around their marker*/
else color='graycc'; /* days without reported events get a light gray border */
text='25a1'x; output; /* this is a border-outline square marker, of the 'albany
amt/unicode' font */
run;

```

In order to be able to trace specific study subjects, we also annotate the participants' ID values to the left of the calendar day markers.

```

data anno_ids; set start_dates;
length function $8 color $12 style $35 text $50;
xsys='2'; ysys='2'; hsys='d'; when='a';
x=start_day; y=y_axis_number;
function='label'; style='albany amt'; position='d'; size=6;
text=trim(left(id))||'a0a0'x; /* these 'a0' hex characters add a little space between
the text and the marker */
run;

```

Per SAS (2010), “the *xsys* variable defines the coordinate system and area of the output used by the X and XC variables to display the Annotate graphics. The *ysys* variable defines the coordinate system and area of the output used by Y and YC to display the Annotate graphics. The *hsys* variable defines the coordinate system and area of the output used by the SIZE variable to display the Annotate graphics. Additionally, you can use the HSYS variable with Java or ActiveX to control the markersize and linesize for the BAR, DRAW, DRAW2TXT, POLY, and SYMBOL functions. The *when* variable specifies when the function is performed in relation to generating other graphics output for the procedure or in relation to generating other Annotate graphics.”

It would also be possible to annotate subject-level descriptive statistics such as mean number of drinks per drinking day, minimum or maximum drinks, or percentage of heavy drinking days to the chart. We would merely have to repeat the above process for the last day of each TLFB period (call that variable *end\_date*) and annotate the descriptive statistics after the *end\_date* marker using a similar DATA step as above.

Since drinking among college students may occur more frequently and with greater intensity on weekends rather than weekdays, we are also including blue reference lines for all calendar days that evaluate to Friday or Saturday, our suspected weekend drinking nights.

```

data anno_weekends;
length function $8 color $12 style $35 text $50;
xsys='2'; ysys='1'; hsys='3'; when='b';
do x='01apr2010'd to '01nov2010'd by 1;
  if trim(left(put(x,downname.))) in ('Friday' 'Saturday') then do;
    function='move'; y=0; output;
    function='draw'; y=100; size=.001; color='cx0276FD'; output;
  end;
end;
run;

```

We now have three data sets that contain items for our final heat map. The data set *anno\_markers* contains all the daily markers that are color-coded by drinking intensity and event (or lack thereof). The data set *anno\_ids* contains the row labels populated with the unique participant ID values. And the third data set is *anno\_weekends* which holds the annotation information for the blue, vertical reference lines for weekend nights (Friday and Saturday).

```

/* anything other than anno_markers goes into anno_other */
data anno_other; set anno_ids anno_weekends;
run;

```

## Creation of the Heat Map using ODS HTML and PROC GPLOT

After we specified our output path and file name above, it is time to put the various graph elements together. Or we could say: “Bring on the heat (map)!”

```

goptions device=png;
goptions xpixels=1800 ypixels=3622;
goptions noborder;

ODS LISTING CLOSE;
ODS HTML path=odsout body="&name..htm"
  (title="Heat Map of Drinking Patterns with Event Occurrence Annotations")
  style=htmlblue;

goptions gunit=pct htitle=14pt ftitle="albany amt/bold" htext=10pt ftext="albany amt";

axis1 label=none value=none order=(1 to &max_y by 1) major=none offset=(.4,.3);

axis2 label=none order=('01apr2010'd to '01nov2010'd by month) major=(height=.2)
offset=(0,0);

symbol value=point color=white; /* use 'invisible' markers (since we're annotating
the real markers) */

title1 justify=left move=(+3,+0) height=14pt "Alcohol Consumption Study";
title2 justify=left move=(+3,+0) height=12pt "Year 2010";

proc gplot data=events3 anno=anno_markers;
format cal_day date9.;
plot y_axis_number*cal_day=1 /
  vaxis=axis1 vreverse
  haxis=axis2
  anno=anno_other
  des='' name="&name";
run;

quit;
ODS HTML CLOSE;
ODS LISTING;


```

The above code creates both a PNG picture file and a .htm file that can be opened in a standard web browser. The top and bottom portions of the large .htm file are displayed below for illustration purposes, but the actual file is larger in size and can be zoomed in or out. Notice that each participant's drinking trajectory is displayed in this heat map, rather than reducing the information down to summary values. Event blocks of particular relevance to the student body in this study stick out as blue "pillars," indicating that most students deemed these events to be helpful anchors.

Several associations are readily available from the heat map. Drinking intensity is higher on suspected weekend nights (Friday or Saturday) than on days preceding school (Sunday through Thursday). Certain events also seem strongly associated with higher drinking intensity.

## Alcohol Consumption Study

Year 2010



[...]






Figure 3: Heat Map of 447 drinking trajectories over time

### BOXPLOTS OF STANDARD DRINKS BY CALENDAR DAY

From our heat map above, we can draw some preliminary conclusions about the association of reported events with drinking intensity. For example, upon inspection of the entire map, it appears that drinking intensity is higher on the fourth and fifth weekend of May 2010. In order to display a chart that summarizes drinking intensity by calendar day, we use PROC BOXPLOT to create a vertical boxplot of drinking amounts for each day of the study period:

```
proc sort data=events3;
by Cal_Day;
run;
title "Box Plots for Standard Drinks by Calendar Day";
proc boxplot data=events3;
plot StDrinks*Cal_Day;
run;
title;
```




Figure 4: One of 9 graphs produced by PROC BOXPLOT

The median drink count as well as interquartile ranges and extreme data values seem to spike on May 21-22 and

May 28-29. Interestingly enough, those are the two weekends (Friday and Saturday) following the end of final exam week of the Spring 2010 semester as well as the beginning of Memorial Day weekend. Clearly, alcohol consumption was higher on those days than the rest in this example with the exception of May 8 (which might either indicate heavier drinking "post-Cinco de Mayo" or, half-jokingly, celebrating Mother's Day, maybe?).

## DISCUSSION

We used grouped time series plots, a heat map, and vertical boxplots to visually inspect our longitudinal drinking data for patterns and associations with both event categories as well as biomarker values. Based on preliminary findings, we have reason to believe that there are associations between social context and drinking frequency and intensity that warrant further statistical analysis.

Study limitations include the use of non-missing data values and there are concerns that pre-filled event anchors might bias the association between drinking and social context as participants could passively retain certain event anchors on their calendars that they may otherwise not have actively recalled. Regarding the biomarker values in hair and fingernails, there is a risk for contamination of the physical specimens by chemical products (e.g. hair dye or perming), nail polish remover or hand lotions containing alcohol.

## CONCLUSION

Time series plots are useful to illustrate longitudinal drinking data when the number of subjects included in the study is not large. Grouped comparisons using penalized B-splines can show differences in trends over time. Large heat maps produced by annotations in PROC GPLOT are valuable tools to detect interesting trends in drinking behavior. Vertical boxplots of daily drinking intensity can be helpful to drill down to specific days of interest as identified after inspection of the heat map.

Future work will make use of event categories to test whether statistically significant associations between social context and drinking behavior exist in our data. Work on a complete dashboard that produces all graphics presented in this paper in a single template is under way.

## ACKNOWLEDGEMENTS

This work was sponsored by the National Institute on Alcohol Abuse and Alcoholism (NIAAA) grant R44AA016463. The funders did not play any role in study design, data collection and analysis, publication decisions, or the preparation of this manuscript.

The first author would like to express his sincere gratitude to Robert Allison from the SAS Institute, Inc. for providing a draft program of the heat map visualization. Mr. Allison's examples of data visualizations on his website [www.robslink.com](http://www.robslink.com) are inspiration and motivation alike.

## REFERENCES

- Beck, K.H., Arria, A.M., Caldeira, K.M., Vincent, K.B., O'Grady, K.E., Wish, E.D. (2008). Social Context of Drinking and Alcohol Problems Among College Students. *Am J Health Behav*, 32(4), 420–430.
- Berger, L., Fendrich, M., Jones, J., Fuhrmann, D., Plate, C., Lewis, D. (2013). Ethyl glucuronide in hair and fingernails as a long-term alcohol biomarker. *Addiction*, 109(3), 425-431.
- Center for Psychological Studies (CPS). Excel Alcohol TLFB (PC Version). Downloaded on 08/24/2014 from: [http://www.nova.edu/gsc/online\\_files.html](http://www.nova.edu/gsc/online_files.html)
- Jones, J., Jones, M., Plate, C., Lewis, D., Fendrich, M., Berger, L., Fuhrmann, D. (2012). Liquid Chromatography-Tandem Mass Spectrometry Assay to Detect Ethyl Glucuronide in Human Fingernail: Comparison to Hair and Gender Differences. *American Journal of Analytical Chemistry*, 3, 83-91.
- Krentzman, A. R., Robinson, E. A., Jester, J. M., Perron, B.E. (2011). Heat maps: a technique for classifying and analyzing drinking behavior. *Substance Use & Misuse*, 46(5), 687–695.
- SAS Institute Inc. (2010). SAS/GRAPH® 9.2 Reference, Second Edition. Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2014). SAS® 9.4 ODS Graphics: Procedures Guide, Third Edition. Cary, NC: SAS Institute Inc.
- Sobell, L. C., Brown, J., Leo, G. I., & Sobell, M. B. (1996). The reliability of the alcohol timeline follow-back when administered by telephone and by computer. *Drug and Alcohol Dependence*, 42(1), 49–54.

Sobell, L. C., & Sobell, M. D. (1992). Timeline follow-back: A technique for assessing self-reported alcohol consumption. In R. Litten & J. Allen (Eds.), *Measuring alcohol consumption: Psychosocial and biochemical methods* (pp. 41–72). Totowa, NJ: Humana Press.

Society of Hair Testing (2011). Consensus of the Society of Hair Testing on Hair Testing for Chronic Excessive Alcohol Consumption 2011.

<http://soht.org>

## **SAS® MACRO RESOURCES**

None.

## **CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Daniel Fuhrmann, Ph.D.  
University of Wisconsin-Milwaukee  
Center for Applied Behavioral Health Research  
P.O. Box 786  
Milwaukee, WI 53201  
Work Phone: (414) 229-5176  
Email: [fuhrmann@uwm.edu](mailto:fuhrmann@uwm.edu)  
Web: <http://www4.uwm.edu/cabhr/>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.